
.SUBCKT OP484_BOYLE 1 2 99 50 45

* Node assignments  | | |  |  |

*                   | | |  |  |

*                   | | |  |  output

*                   | | |  negative supply

*                   | | positive supply

*                   | inverting input

*                   non-inverting input         

*

* This version of the OP484 model is a Boyle Model of the OP484E

* versus the Analog Devices Supplied MPZ Model. The AD supplied

* model is used for the input stage and a two pole/one zero 

* with constant output impedance output stage has been impemented.  

* The AD derived output stage was not used because closed loop 

* operation is well below the unity gain bw of the system and the 

* output is lightly loaded.

*

* Clint Brown Aeroflex Labs 8/24/04

*

*

*INPUT STAGE

*

Q_Q1              5 2 3 QIN

Q_Q2              6 11 3 QIN

Q_Q3              7 2 4 QIP

Q_Q4              8 11 4 QIP

D_DC1             2 11 DC 

D_DC2             11 2 DC 

Q_Q5              4 9 99 QIP

Q_Q6              9 9 99 QIP

Q_Q7              3 10 50 QIN

Q_Q8              10 10 50 QIN

R_R1              5 99  4k  

R_R2              6 99  4k  

R_R3              50 7  4k  

R_R4              50 8  4k  

I_IREF            9 10 DC 50.5e-6  

E_EOS             1 11 POLY(2) (22,98) (14,98) -25E-6 1E-2 1

I_IOS             2 1 DC 5E-9  

C_CIN             1 2  2P  

G_GIN1            98 1 VALUE { (V(23) -V(98)) *.001    }

G_GIN2            98 2 VALUE { (V(17) -V(98)) *.001    }

*

*VOLTAGE NOISE SOURCE WITH FLICKER NOISE

*

D_VN1             13 98 DC 2 

D_VN2             98 15 DC 2 

D_DN1             13 14 DEN 

D_DN2             14 15 DEN 

*

*CURRENT NOISE SOURCE WITH FLICKER NOISE

*

D_VN3             16 98 DC 2 

D_VN4             98 18 DC 2 

D_DN3             16 17 DIN 

D_DN4             17 18 DIN 

*

*2ND CURRENT NOISE SOURCE WITH FLICKER NOISE

*

D_VN5             19 98 DC 2 

D_VN6             98 24 DC 2 

D_DN5             19 23 DIN 

D_DN6             23 24 DIN 

*

*FALSE GROUND STAGE

*

R_RCM1            97 99  10k 

R_RCM2            50 97  10k 

E_GAIN98          98 0 VALUE {1 * V(97)}

*

*COMMON MODE STAGE WITH ZERO AT 100HZ

*

E_ECM             98 21 POLY(2) (1,98) (2,98) 0 .5 .5

R_R10             21 22  1K  

C_C4              22 21  1.592E-6  

R_R11             98 22  .1  

*

*GAIN STAGE WITH POLE AT 32HZ AND CLAMPING AT SUPPLY RAILS

*

G_GAIN1N          98 30 VALUE { (V(5) -V(6))*.0025    }

G_GAIN1P          98 30 VALUE { (V(7) -V(8))*.0025    }

R_Rgain1          98 30  20E6 

C_Cpole1          98 30  25e-11  

D_DC3             31 30 DC

D_DC4             30 32 DC

V_V1              31 50 .8V

V_V2              99 32 .8V

*

*UNITY GAIN STAGE WITH POLE AT 800 KHZ AND ZERO AT 5MHZ 

*

G_GAIN2           98 33 VALUE { (V(30) -V(98))*.001    }

R_Rzero1          34 33  150  

C_Cpole2          98 34  200P  

R_Rgain2          98 33  1K  

*

* OUTPUT STAGE

*

E_GAIN3           35 0 VALUE {1 * V(33)}

R_Rout            35 45  35  

*

.MODEL DC D(IS=130E-21)

.MODEL DEN D(RS=100 KF=12E-15 AF=1)

.MODEL DIN D(RS=5.358 KF=56E-15 AF=1)

.MODEL QIN NPN(BF=250 VA=200 IS=0.5E-16)

.MODEL QIP PNP(BF=60 VA=100 IS=0.5E-16)

.ENDS   OP484_BOYLE

*$

